Table 2.1 Context Free Syntax of Mini-Language Core in BNF

```
<program>               ::=  program
                               <declaration-sequence>
                             begin
                               <statement-sequence>
                             end;

<declaration-sequence>  ::=  <declaration>
                           |  <declaration> <declaration-sequence>

<statement-sequence>    ::=  <statement>
                           |  <statement> <statement-sequence>

<declaration>           ::=  declare  <identifier-list>;

<identifier-list>       ::=  <identifier>
                           |  <identifier>, <identifier-list>

<statement>             ::=  <assignment-statement>
                           |  <if-statement>
                           |  <loop-statement>
                           |  <input-statement>
                           |  <output-statement>

<assignment-statement>  ::=  <identifier> := <expression> ;

<if-statement>          ::=  if  <comparison> then
                               <statement-sequence>
                             end if;
                           |  if <comparison> then
                               <statement-sequence>
                             else
                               <statement-sequence>
                             end if;

<loop-statement>        ::=  while <comparison> loop
                               <statement-sequence>
                             end loop;

<input-statement>       ::=  input  <identifier-list> ;

<output-statement>      ::=  output <identifier-list> ;
```

Table 2.1 continued

<comparison>	::=	(<operand> = <operand>)
	|	(<operand> ≠ <operand>)
	|	(<operand> ≤ <operand>)
	|	(<operand> ≥ <operand>)
<expression>	::=	<factor>
	|	<expression> + <factor>
	|	<expression> - <factor>
<factor>	::=	<operand>
	|	<factor> * <operand>
<operand>	::=	<integer>
	|	<identifier>
	|	(<expression>)
<identifier>	::=	<letter>
	|	<identifier> <letter>
	|	<identifier> _ <letter>
<integer>	::=	<digit> | <integer> <digit>
<letter>	::=	A | B | C | D | E | F | G | H | I
	|	J | K | L | M | N | O | P | Q | R
	|	S | T | U | V | W | X | Y | Z
<digit>	::=	0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

Table 2.4 Syntax of Mini-language Core Defined by Syntax Charts

PROGRAM



DECLARATION



STATEMENT



EXPRESSION



FACTOR



Table 2.4 continued